Concurrent Erlang

Erlang SOLUTIONS

Overview: concurrent Erlang

- Creating Processes
- Message Passing
- Receiving Messages
- Data in Messages
- Registered Processes
- Timeouts
- More on Processes
- Observer Processes

The Road to 2 Million Websocket Connections in Phoenix

Posted on November 3rd, 2015 by Gary Rennie





If you have been paying attention on Twitter recently, you have likely seen some increasing numbers regarding the number of simultaneous connections the Phoenix web framework can handle. This post documents some of the techniques used to perform the benchmarks.

Concurrency vs Parallelism



- Concurrency happens when your code is running in different processes
- Control of Concurrency is key to scale
- Concurrent solutions can exploit the underlying system's parallelism, if present
- Parallelism can speed up execution

Erlang SOLUTIONS

Creating Processes

```
Pid2 = spawn(Mod, Func, Args)
```



- ▶️ Before
  - Code executed by Process 1
  - **process identifier** is Pid1
  - Pid2 = spawn(M, F, A)
- ▶️ After
  - A new process with Pid2 is created
  - Pid2 is only known to Pid1
  - Pid2 runs M:F(A)
  - M:F/Arity must be exported
- ▶️ Convention: we identify processes by their process ids (pids)

Creating Processes

```
Pid2 = spawn(Mod, Func, Args)
```



- The BIF **spawn** fails when max number of running processes reached
- A process terminates
  - **abnormally** when run-time errors occur
  - **normally** when there is no more code to execute

Erlang SOLUTIONS

Message Passing




```
Pid2 ! {self(),foo}
```

- Messages are sent using the **Pid ! Msg** expression
  - **Msg** is any valid Erlang data type
- Sending a message will never fail
- Messages sent to non-existing processes are thrown away
- Received messages are stored in the process' mailbox

Message Passing



Receiving Messages

```
receive
    Pattern1 ->
        <expression 1>,
        <expression 2>,
        ...
        <expression N>;
    Pattern2 ->
        <expression 1>,
        ...
        <expression N>;
    PatternN ->
        <expression 1>,
        ...
        <expression N>
end
```

- Messages are retrieved using a **receive** clause
- **receive** suspends the process until a matching message is received
- Message passing is asynchronous

Receiving Messages




```
Pid2 ! {self(),foo}
```

```
receive
  start      -> start_it();
  stop       -> stop_it();
  {Pid, foo} -> foo(Pid)
end
```

- Messages can be matched and selectively retrieved
- Messages are received when a message matches a clause
- Mailboxes are scanned sequentially.

Receiving Messages




```
receive
  {Pid,{digit,Digit}} ->
     digit(Pid, Digit)
end
```

- If **Pid** is bound before receiving the message, then only data tagged with that pid can be pattern matched
- The variable **Digit** is bound when receiving the message

Receiving Messages: selective



- The message **foo** is received, followed by the message **bar**
- This is irrespective of the order in which they were sent or stored in the mailbox

```
receive
  foo -> true
end,
receive
  bar -> true
end,
...
```

Receiving Messages: non-selective



- The first message to arrive at the process **Pid3** will be processed
- The variable **Msg** in the process **Pid3** will be bound to one of the atoms **foo** or **bar** depending on which arrives first.

```
receive
  Msg -> true
end,
...
```

Receiving Messages



- PidA sends a message to PidB containing its own Pid
- PidB binds it to variable Pid and sends a message to PidC
- PidC receives the message and replies directly to PidA

Data in Messages: example




```
-module(echo).
-export([go/0, loop/0]).

go() ->
  Pid = spawn(echo,loop,[]),
  Pid ! {self(), hello},
  receive
    {Pid, Msg} ->
      io:format("~w~n",[Msg])
  end,
  Pid ! stop.
```

```

loop() ->
  receive
    {From, Msg} ->
      From ! {self(), Msg},
      loop();
    stop ->
      ok
  end.
```

Registered Processes

```
register(Alias, Pid)
Alias ! Message
```

- Registers the process **Pid** with the name **Alias**, which has to be an atom
- Any process can send a message to a registered process
- The BIF **registered/0** returns all registered process names
- The BIF **whereis(Alias)** returns the Pid of the process with the name **Alias**.

Message Passing

Pid1



```
does_not_exist ! hello
```

- Sending messages to non-existing registered processes causes the calling process to terminate with a **badarg** error

Registered Processes




```
echo ! {self(), hello}
```

```
-module(echo).
-export([go/0, loop/0]).

go() ->
  Pid = spawn(echo,loop,[]),
  register(echo, Pid).
```

```
loop() ->
  receive
    {From, Msg} ->
      From ! {self(), Msg},
      loop();
    stop ->
      ok
  end.
```

Timeouts




```
receive
  Msg ->
    <expressions1>
after TimeOut ->
    <expressions2>
end
```

- If the message **Msg** is received within the time **TimeOut**, <expressions1> will be executed
- If not, <expressions2> will be executed
- TimeOut is an integer denoting the time in milliseconds or the atom **infinity**

Timeouts

```
read(Key) ->
    db ! {self(), {read, Key}},
    receive
      {read, R} ->
         {ok, R};
      {error, Reason} ->
         {error, Reason}
    after 1000 ->
      {error, timeout}
    end.
```

- If the server takes more than a second to handle the request, a timeout is generated
- Do not forget to handle messages received after a timeout

Timeouts

```
send_after(Time, Msg) ->
    spawn(timer,
          send,
          [self(),Time,Msg]).

send(Pid, Time, Msg) ->
    receive
    after Time ->
        Pid ! Msg
    end.

sleep(T) ->
    receive
    after T ->
        true
    end.
```

- **send_after(T, What)** sends the message **What** to the current process after **T** milliseconds
- The **sleep(T)** function will suspend the calling process for **T** milliseconds

Timeouts

```
flush() ->
  receive
    _ -> flush()
  after 0 ->
    ok
  end.
```

- **flush()** will clear the mailbox from all messages, stopping when it is empty.

More on Processes: definitions

Process

A concurrent activity. The system may have many concurrent processes executing at the same time

Message

A method of communication and sharing data between processes

Timeout

A mechanism for waiting for a given period of time for an incoming message

More on Processes: definitions

Registered Processes

Processes which have been given a name with BIFs such as **register/2**.

Termination

A process is said to terminate **normally** when it has no more code to execute.

It terminates **abnormally** if a run time error occurs or if someone makes it exit with a non-normal reason.

More on Processes: process skeleton

```
start(Args) ->
   spawn(server, init, [Args])


init(Args) ->
  State = initialize_state(Args),
  loop(State).


loop(State) ->
   receive
      {handle, Msg} ->
         NewState = handle(Msg, State),
         loop(NewState);
      stop -> terminate(State)
   end.


terminate(State) -> clean_up(State).
```



Observer Processes

- A component of the observer graphical tool used to inspect the state of processes
  - Including processes in connected nodes
- Trace output for messages sent & received
- Trace output for process events such as spawn, exit and link
- Trace output for BIF and function calls
- Replaces the **pman** tool which was discontinued after R16

Observer Processes



➢ observer:start()

Observer Processes



➢ Prints process information

Observer Processes



- Pick what trace messages you want to view
- Pick the inheritance level when spawning

Observer Processes

Trace Log

```
18:10:10:520391 (<0.32.0>) << {shell_cmd,<0.26.0>,
                                         {eval,[{op,1,'+',{integer,1,1},{integer,1,2}}]},
                                         cmd}
18:10:10:520475 (<0.32.0>) <0.25.0> ! {io_request,<0.32.0>,<0.25.0>,getopts}
18:10:10:520497 (<0.32.0>) << {io_reply,<0.25.0>,
                                        [{expand_fun,#Fun<group.0.100149429>},
                                         {echo,true},
                                         {binary,false},
                                         {encoding,unicode}]}
18:10:10:520515 (<0.32.0>) <0.25.0> ! {io_request,<0.32.0>,<0.25.0>,{get_geometry,columns}}
18:10:10:520537 (<0.32.0>) << {io_reply,<0.25.0>,80}
18:10:10:520552 (<0.32.0>) <0.25.0> ! {io_request,<0.32.0>,<0.25.0>,
                                          {requests,
                                              [{put_chars,unicode,<<"3">>},
                                               {put_chars,unicode,"\n"}]}}
18:10:10:520565 (<0.32.0>) << {io_reply,<0.25.0>,ok}
18:10:10:520592 (<0.32.0>) <0.26.0> ! {shell_rep,<0.32.0>,{value,3,[],[]}}
```

- Start tracing in the Trace Overview
- Prints trace information

Summary: concurrent Erlang

- Creating Processes
- Message Passing
- Receiving Messages
- Data in Messages
- Registered Processes
- Timeouts
- More on Processes
- Observer Processes